# Kleinste-kwadraten

Academiejaar 2020-2021

## Systeem

```
> restart;
> with( LinearAlgebra ):
  with(plots):
  interface( rtablesize = 16 ):
```

**Opgave:** In de onderstaande tabel staan de data van een denkbeeldig scheikundig experiment. Bereken eerst de interpolerende veelterm van graad 15 door de gegeven 16 knooppunten. Lijkt dit een goede fit? Bereken vervolgens de kleinste-kwadraten veelterm van graad 5 voor de gegeven knooppunten. Geeft dit wel een gedrag dat je zou verwachten? Geef en bespreek in beide gevallen het conditiegetal. Kan je dit nog verbeteren?

| **x-data** | 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9 | 10 | 11 | 12 | 13 | 14 | 15 |
|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|
| **y-data** | 12 | 20 | 22 | 20 | 16 | 12 | 8.5 | 6 | 4.5 | 4.5 | 5.5 | 6.5 | 7.5 | 8.5 | 9 | 9 |

```
> x := [seq(i,i=0..15)];
  y := [12, 20, 22, 20, 16, 12, 17/2, 6, 9/2, 9/2, 11/2, 13/2, 15/2,
  17/2, 9, 9];
```

$$x := [0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15]$$

$$y := \left[12, 20, 22, 20, 16, 12, \frac{17}{2}, 6, \frac{9}{2}, \frac{9}{2}, \frac{11}{2}, \frac{13}{2}, \frac{15}{2}, \frac{17}{2}, 9, 9\right] \quad \textbf{(1)}$$

```
> p1 := pointplot(x,y,symbol=solidcircle,symbolsize=10,color=red,
  size=[600,400],legend="datapunten");
```




## 3.1. Veelterminterpolatie

We kunnen een veelterminterpolant ook berekenen aan de hand van een lineair stelsel (in het algemaan niet aangeraden door de slechte conditionering), we hebben namelijk de inerpolatievoorwaarden

$$\sum_{i=0}^{15} c_i \, x_k^{\;i} = y_i \qquad k = 0, 1, \ldots, 15$$

Als we dit in maxtrix-vector notatie uitschrijven (Ac=b) vinden we een Vandermonde matrix terug als coëfficiëntenmatrix.

```
> A := Matrix(16,16,(i,j)->x[i]^(j-1));
```

**(2.1)**

$A :=$ [[1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0], **(2.1)**
[1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1],
[1, 2, 4, 8, 16, 32, 64, 128, 256, 512, 1024, 2048, 4096, 8192, 16384, 32768],
[1, 3, 9, 27, 81, 243, 729, 2187, 6561, 19683, 59049, 177147, 531441, 1594323,
4782969, 14348907],
[1, 4, 16, 64, 256, 1024, 4096, 16384, 65536, 262144, 1048576, 4194304,
16777216, 67108864, 268435456, 1073741824],
[1, 5, 25, 125, 625, 3125, 15625, 78125, 390625, 1953125, 9765625, 48828125,
244140625, 1220703125, 6103515625, 30517578125],
[1, 6, 36, 216, 1296, 7776, 46656, 279936, 1679616, 10077696, 60466176,
362797056, 2176782336, 13060694016, 78364164096, 470184984576],
[1, 7, 49, 343, 2401, 16807, 117649, 823543, 5764801, 40353607, 282475249,
1977326743, 13841287201, 96889010407, 678223072849, 4747561509943],
[1, 8, 64, 512, 4096, 32768, 262144, 2097152, 16777216, 134217728,
1073741824, 8589934592, 68719476736, 549755813888, 4398046511104,
35184372088832],
[1, 9, 81, 729, 6561, 59049, 531441, 4782969, 43046721, 387420489,
3486784401, 31381059609, 282429536481, 2541865828329, 22876792454961,
205891132094649],
[1, 10, 100, 1000, 10000, 100000, 1000000, 10000000, 100000000, 1000000000,
10000000000, 100000000000, 1000000000000, 10000000000000,
100000000000000, 1000000000000000],
[1, 11, 121, 1331, 14641, 161051, 1771561, 19487171, 214358881, 2357947691,
25937424601, 285311670611, 3138428376721, 34522712143931,
379749833583241, 4177248169415651],
[1, 12, 144, 1728, 20736, 248832, 2985984, 35831808, 429981696, 5159780352,
61917364224, 743008370688, 8916100448256, 106993205379072,
1283918464548864, 15407021574586368],
[1, 13, 169, 2197, 28561, 371293, 4826809, 62748517, 815730721, 10604499373,
137858491849, 1792160394037, 23298085122481, 302875106592253,
3937376385699289, 51185893014090757],
[1, 14, 196, 2744, 38416, 537824, 7529536, 105413504, 1475789056,
20661046784, 289254654976, 4049565169664, 56693912375296,
793714773254144, 11112006825558016, 155568095557812224],
[1, 15, 225, 3375, 50625, 759375, 11390625, 170859375, 2562890625,
38443359375, 576650390625, 8649755859375, 129746337890625,
1946195068359375, 29192926025390625, 437893890380859375]]

```
> b := Vector(16,i -> y[i]);
```

$$b := \begin{bmatrix} 12 \\ 20 \\ 22 \\ 20 \\ 16 \\ 12 \\ \frac{17}{2} \\ 6 \\ \frac{9}{2} \\ \frac{9}{2} \\ \frac{11}{2} \\ \frac{13}{2} \\ \frac{15}{2} \\ \frac{17}{2} \\ 9 \\ 9 \end{bmatrix} \tag{2.2}$$

```
> print(kappa('A') = evalf(ConditionNumber(A)));
```

$$\kappa(A) = 9.368457901\ 10^{21} \tag{2.3}$$

```
> c := LinearSolve(A,b,method='LU');
```

$$c := \begin{bmatrix} 12 \\ \frac{77377}{624} \\ -\frac{107776461679}{302702400} \\ \frac{234725267863}{504504000} \\ -\frac{1306582489}{3742200} \\ \frac{50500687}{299376} \\ -\frac{609504061}{10886400} \\ \frac{2887016327}{217728000} \\ -\frac{2791253743}{1219276800} \\ \frac{4728281}{16257024} \\ -\frac{4730981}{174182400} \\ \frac{1603517}{870912000} \\ -\frac{168793}{1916006400} \\ \frac{2809}{996323328} \\ -\frac{9439}{174356582400} \\ \frac{137}{290594304000} \end{bmatrix} \tag{2.4}$$

```
> p2 := plot(add(c[i]*X^(i-1),i=1..16),X=0..15,color=blue,legend=
  "Veelterminterpolatie",thickness=2):
  display(p2,p1);
```




We merken direct de 'vreemde' bulten op in de buurt van de randen van het interval. Uit het hoofdstuk over interpolatie weten we dat dit een gekend fenomeen is voor interpolatie met veeltermen van hogere graad.

## 5.1. en 5.2. Overbepaalde stelsels

We kunnen de data niet goed fitten met veelterminterpolatie doordat de veelterm van een veel te hoge graad is en bijgevolg oscilleert. Wat als we nu een veelterm van lagere graad zouden proberen te fitten?

```
> unassign('A','b','c'):
  n := 5:
> A := Matrix(16,n+1,(i,j)->x[i]^(j-1));
```

$$A := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 1 & 1 & 1 & 1 & 1 & 1 \\ 1 & 2 & 4 & 8 & 16 & 32 \\ 1 & 3 & 9 & 27 & 81 & 243 \\ 1 & 4 & 16 & 64 & 256 & 1024 \\ 1 & 5 & 25 & 125 & 625 & 3125 \\ 1 & 6 & 36 & 216 & 1296 & 7776 \\ 1 & 7 & 49 & 343 & 2401 & 16807 \\ 1 & 8 & 64 & 512 & 4096 & 32768 \\ 1 & 9 & 81 & 729 & 6561 & 59049 \\ 1 & 10 & 100 & 1000 & 10000 & 100000 \\ 1 & 11 & 121 & 1331 & 14641 & 161051 \\ 1 & 12 & 144 & 1728 & 20736 & 248832 \\ 1 & 13 & 169 & 2197 & 28561 & 371293 \\ 1 & 14 & 196 & 2744 & 38416 & 537824 \\ 1 & 15 & 225 & 3375 & 50625 & 759375 \end{bmatrix} \quad \textbf{(3.1)}$$

```
> b := Vector(16,i -> y[i]);
```

**(3.2)**

$$b := \begin{bmatrix} 12 \\ 20 \\ 22 \\ 20 \\ 16 \\ 12 \\ \frac{17}{2} \\ 6 \\ \frac{9}{2} \\ \frac{9}{2} \\ \frac{11}{2} \\ \frac{13}{2} \\ \frac{15}{2} \\ \frac{17}{2} \\ 9 \\ 9 \end{bmatrix} \tag{3.2}$$

```
> print( 'A' = A, 'c' = Vector(n+1,i->c[i-1]), 'b' = b );
```

(3.3)

$$A = \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 1 & 1 & 1 & 1 & 1 & 1 \\ 1 & 2 & 4 & 8 & 16 & 32 \\ 1 & 3 & 9 & 27 & 81 & 243 \\ 1 & 4 & 16 & 64 & 256 & 1024 \\ 1 & 5 & 25 & 125 & 625 & 3125 \\ 1 & 6 & 36 & 216 & 1296 & 7776 \\ 1 & 7 & 49 & 343 & 2401 & 16807 \\ 1 & 8 & 64 & 512 & 4096 & 32768 \\ 1 & 9 & 81 & 729 & 6561 & 59049 \\ 1 & 10 & 100 & 1000 & 10000 & 100000 \\ 1 & 11 & 121 & 1331 & 14641 & 161051 \\ 1 & 12 & 144 & 1728 & 20736 & 248832 \\ 1 & 13 & 169 & 2197 & 28561 & 371293 \\ 1 & 14 & 196 & 2744 & 38416 & 537824 \\ 1 & 15 & 225 & 3375 & 50625 & 759375 \end{bmatrix}, c = \begin{bmatrix} c_0 \\ c_1 \\ c_2 \\ c_3 \\ c_4 \\ c_5 \end{bmatrix}, b = \begin{bmatrix} 12 \\ 20 \\ 22 \\ 20 \\ 16 \\ 12 \\ \frac{17}{2} \\ 6 \\ \frac{9}{2} \\ \frac{9}{2} \\ \frac{11}{2} \\ \frac{13}{2} \\ \frac{15}{2} \\ \frac{17}{2} \\ 9 \\ 9 \end{bmatrix} \tag{3.3}$$

We merken op dat dit stelsel overbepaald is. Dit wil zeggen dat we meer voorwaarden (aantal rijen = 16) hebben dan onbekende parameters (aantal kolommen = 6). Bijgevolg zal dit stelsel in het algemeen geen exacte oplossing hebben. We moeten dus eerst definiëren wat we bedoelen met het oplossen van dit stelsel.

Aangezien er geen exacte oplossing bestaat, kunnen we verschillende benaderingen het beste vergelijken aan de hand van het residu $r = b - Ac$. Dit is natuurlijk een vector van dezelfde dimensie als het rechterlid $b$ van het lineaire stelsel, dus we gebruiken de norm van deze vector om de kwaliteit van een benadering te kwantificeren. wij zijn in dit geval geïnteresseerd in de 2-norm van het residu:

$$\|r\|_2 = \sum_{i=1}^{n} r_i^2$$

De vector $c$ die de 2-norm van het residu minimaliseert noemen we de kleinste-kwadraten oplossing van het lineaire stelsel. We kunnen natuurlijk ook andere nomen minimaliseren en dit leidt dan tot andere benaderingen met verschillende eigenschappen, maar dit behoort verder niet tot het materiaal van deze cursus.

We kunnen aantonen dat de oplossing van dit minimalisatieprobleem gelijk is aan de oplossing van het vierkante stelsel van normaalvergelijkingen, die verkregen worden door het voormenigvuldigen van het overbepaald stelsel met de getransponeerde van de coëfficiëntenmatrix:

$Ac = b \quad => \quad A^T A\, c = A^T b$

Dit stelsel is nu vierkant en kan makkelijk opgelost worden aan de hand van zijn LU-decompositie.

```
> print( 'A^T*A' = Transpose(A).A, 'c' = Vector(n+1,i->c[i-1]),
  'A^T*b' = Transpose(A).b );
```

$$A^T A = [[16, 120, 1240, 14400, 178312, 2299200], [120, 1240, 14400, 178312, 2299200, 30482920], [1240, 14400, 178312, 2299200, 30482920, 412420800], [14400, 178312, 2299200, 30482920, 412420800, 5666482312], [178312, 2299200, 30482920, 412420800, 5666482312, 78800938560], [2299200, 30482920, 412420800, 5666482312, 78800938560, 1106532668200]], c = \begin{bmatrix} c_0 \\ c_1 \\ c_2 \\ c_3 \\ c_4 \\ c_5 \end{bmatrix}, A^T b = \begin{bmatrix} \frac{343}{2} \\ \frac{2011}{2} \\ \frac{19477}{2} \\ \frac{227191}{2} \\ \frac{2873581}{2} \\ \frac{37866871}{2} \end{bmatrix} \tag{3.4}$$

De conditionering van dit stelsel is echter veel slechter dan van het originele stelsel, het conditiegetal is immers kwadratisch gegroeid!

```
> cond_rect := proc(A)
  local s;
  s := SingularValues(evalf(Transpose(A)));
  return max(s)/min(s);
  end proc:
> print('kappa[2](A)' = cond_rect(A));
  print('kappa[2](A^T*A)' = cond_rect(Transpose(A).A));
```

$$\kappa_2(A) = 1.81523969194531\ 10^6$$
$$\kappa_2\left(A^T A\right) = 3.29511284733196\ 10^{12} \quad \textbf{(3.5)}$$

# 5.5. QR-decompositie

Aangezien voorvermenigvuldigen met de getransponeerde matrix nefast is voor de conditionering van het stelsel zijn we op zoek naar een alternatief. Juist zoals voor vierkante stelsels, kunnen we een rechthoekig stelsel dat in bovendriehoeksvorm is heel simpel oplossen door middel van achterwaartse substitutie.

We zoeken dus een matrix $Q^T$ zodat $Q^T A = R$ waarbij $R$ een rechthoekige bovendriehoeksmatrix is. De matrix $Q$ kan echter niet van eender welke vorm zijn. Aangezien we de 2-norm van het residu willen minimaliseren, willen we natuurlijk niet dat de norm van het residu verandert na vermenigvuldiging met $Q^T$ (anders zou het minimum ook kunnen veranderen). Matrices die na vermenigvuldigen de norm behouden noemen we orthogonale matrices. Dergelijke matrices worden gekenmerkt door de eigenschap

$$Q^T Q = Q\, Q^T = I$$

Met andere woorden: hun inverse is gelijk aan hun getransponeerde.

## 5.4. Ortgonale matrices en Householder reflecties

Een heel bekende familie van orthogonale matrices zijn de Householder reflecties. Deze zijn gedefinieerd als

$$H = \mathrm{I} - \rho\, u\, u^T, \qquad \rho = \frac{2}{\|u\|^2}$$

Voor een kolomvector $u$.

```
> u := <1,2,1,-1>;
```

$$u := \begin{bmatrix} 1 \\ 2 \\ 1 \\ -1 \end{bmatrix} \quad \textbf{(4.1.1)}$$

```
> print('u*u^T' = u.Transpose(u));
```

$$u\,u^T = \begin{bmatrix} 1 & 2 & 1 & -1 \\ 2 & 4 & 2 & -2 \\ 1 & 2 & 1 & -1 \\ -1 & -2 & -1 & 1 \end{bmatrix} \qquad \textbf{(4.1.2)}$$

```
> H := IdentityMatrix(4,4) - 2/Norm(u,2)^2 * u.Transpose(u);
```

$$H := \begin{bmatrix} \frac{5}{7} & -\frac{4}{7} & -\frac{2}{7} & \frac{2}{7} \\ -\frac{4}{7} & -\frac{1}{7} & -\frac{4}{7} & \frac{4}{7} \\ -\frac{2}{7} & -\frac{4}{7} & \frac{5}{7} & \frac{2}{7} \\ \frac{2}{7} & \frac{4}{7} & \frac{2}{7} & \frac{5}{7} \end{bmatrix} \qquad \textbf{(4.1.3)}$$

```
> H := HouseholderMatrix(u);
```

$$H := \begin{bmatrix} \frac{5}{7} & -\frac{4}{7} & -\frac{2}{7} & \frac{2}{7} \\ -\frac{4}{7} & -\frac{1}{7} & -\frac{4}{7} & \frac{4}{7} \\ -\frac{2}{7} & -\frac{4}{7} & \frac{5}{7} & \frac{2}{7} \\ \frac{2}{7} & \frac{4}{7} & \frac{2}{7} & \frac{5}{7} \end{bmatrix} \qquad \textbf{(4.1.4)}$$

Deze matrices voldoen aan de volgende eigenschappen:

- $H^T = H$
- $H^T H = H^2 = I$

Bijgevolg zijn de Householder reflecties dus orthogonale matrices.

```
> print('H' = H,'H^T' = Transpose(H));
```

$$H = \begin{bmatrix} \frac{5}{7} & -\frac{4}{7} & -\frac{2}{7} & \frac{2}{7} \\ -\frac{4}{7} & -\frac{1}{7} & -\frac{4}{7} & \frac{4}{7} \\ -\frac{2}{7} & -\frac{4}{7} & \frac{5}{7} & \frac{2}{7} \\ \frac{2}{7} & \frac{4}{7} & \frac{2}{7} & \frac{5}{7} \end{bmatrix}, H^T = \begin{bmatrix} \frac{5}{7} & -\frac{4}{7} & -\frac{2}{7} & \frac{2}{7} \\ -\frac{4}{7} & -\frac{1}{7} & -\frac{4}{7} & \frac{4}{7} \\ -\frac{2}{7} & -\frac{4}{7} & \frac{5}{7} & \frac{2}{7} \\ \frac{2}{7} & \frac{4}{7} & \frac{2}{7} & \frac{5}{7} \end{bmatrix} \qquad \textbf{(4.1.5)}$$

```
> print('H^T*H' = Transpose(H).H);
```

$$H^T H = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{4.1.6}$$

Maar wat doet een vermenigvuldiging met zo'n matrix eigenlijk? We bekijken een 2D voorbeeld.

```
> u := <sqrt(3),1>:
  H := HouseholderMatrix(u):
  uT := 2*<-1,sqrt(3)>:
  x := <3,-1>:
  Hx := H.x:
  y := <sqrt(3),3>:
  Hy := H.y:
  a1 := arrow(x,shape=arrow,color=blue,head_length=0.1,
  head_width=0.1):
  a2 := arrow(u,shape=arrow,color=black,head_length=0.1,
  head_width=0.1):
  a3 := arrow(uT,shape=arrow,color=black,head_length=0,
  head_width=0):
  a4 := arrow(-uT,shape=arrow,color=black,head_length=0,
  head_width=0):
  a5 := arrow(H.x,shape=arrow,color=blue,head_length=0.1,
  head_width=0.1):
  a6 := arrow(y,shape=arrow,color=red,head_length=0.1,
  head_width=0.1):
  a7 := arrow(H.y,shape=arrow,color=red,head_length=0.1,
  head_width=0.1):
  t1 := textplot([x[1],x[2],'x'],align=RIGHT,color=blue):
  t2 := textplot([Hx[1],Hx[2],'Hx'],align=BELOW,color=blue):
  t3 := textplot([u[1],u[2],'u'],align=RIGHT):
  t4 := textplot([uT[1]+0.25,uT[2]-0.1,"&bot;"],align=ABOVE):
  t5 := textplot([uT[1]+0.1,uT[2],"u"],align=ABOVE):
  t6 := textplot([y[1],y[2],'y'],align=ABOVE,color=red):
  t7 := textplot([Hy[1],Hy[2],'Hy'],align=LEFT,color=red):
  display([a1,a2,a3,a4,a5,a6,a7,t1,t2,t3,t4,t5,t6,t7]);
```




We merken op dat de vermenigvuldiging met een Householder matrix eigenlijk een spiegeling is ten opzichte van de rechte loodrecht op $u$. Dit betekent ook dat de lengte van vectoren (oftewel de norm) dus ook bewaard blijft, wat natuurlijk de eigenschap is die we wilden hebben.

Bovendien merken we op dat de vector $y$ wordt afgebeeld op één van de assen, oftewel enkel 1 van de componenten is niet 0. Bijgevolg kunnen we $u$ zo kiezen dat we alle componenten, met eentje uitgezonderd, op 0 kunnen afbeelden. Deze eigenschap zal ons in staat stellen om een matrix te transformeren tot bovendriehoeksvorm.

```
> x := <-1,-2,2,4>;
```

$$x := \begin{bmatrix} -1 \\ -2 \\ 2 \\ 4 \end{bmatrix} \quad \textbf{(4.1.7)}$$

```
> u := x + sign(x[1])* Norm(x,2)*UnitVector(1,4);
```

**(4.1.8)**

$$u := \begin{bmatrix} -6 \\ -2 \\ 2 \\ 4 \end{bmatrix} \tag{4.1.8}$$

```
> H := HouseholderMatrix(u);
```

$$H := \begin{bmatrix} -\frac{1}{5} & -\frac{2}{5} & \frac{2}{5} & \frac{4}{5} \\ -\frac{2}{5} & \frac{13}{15} & \frac{2}{15} & \frac{4}{15} \\ \frac{2}{5} & \frac{2}{15} & \frac{13}{15} & -\frac{4}{15} \\ \frac{4}{5} & \frac{4}{15} & -\frac{4}{15} & \frac{7}{15} \end{bmatrix} \tag{4.1.9}$$

```
> print('H*x' = H.x);
```

$$H\,x = \begin{bmatrix} 5 \\ 0 \\ 0 \\ 0 \end{bmatrix} \tag{4.1.10}$$

## 5.5. QR-decompositie

We gaan nu proberen om de orthogonale matrix Q op te stellen zodat $A = QR$, met $R$ een rechthoekige bovendriehoeksmatrix. Deze matrix $Q$ zal bestaan uit het product van enkele Householdermatrices (in combinatie met eenheidsmatrices), die zo gekozen zijn om alle elementen onder de diagonaal gelijk te stellen aan 0. We gaan van links naar rechts te werk on zo de eerder gecreëerde nullen niet terug teniet te doen.

```
> QR_decomposition := proc(A)
  local m,n,v,H, M, QT,Q,R,k;
  m := RowDimension(A);
  n := ColumnDimension(A);
  R := A;
  QT := IdentityMatrix(m,m);

  print();
  print();
  print();
  print();
  print('kolom' = 1);
  print('R[0]' = R);

  k:= 1;
  v := Vector(m-k+1,i->A[k+i-1,k]);
  H := simplify(HouseholderMatrix(v + sign(v[1])* Norm(v,2)*
```

```
UnitVector(1,m-k+1)));
R := simplify(H.R);
QT := simplify(H.QT);

print();
print('v' = v, 'H' = H, 'M' = H);
print('R[1]' = R);

for k from 2 to n do
   print();
   print();
   print();
   print();
   print('kolom' = k);
   print('R'[k-1] = R);
   v := Vector(m-k+1,i->R[k+i-1,k]);
   H := simplify(HouseholderMatrix(v + sign(evalf(v[1]))* Norm
(v,2)*UnitVector(1,m-k+1)));
   M := <<IdentityMatrix(k-1,k-1)| Matrix(k-1,m-k+1)>,<Matrix
(m-k+1,k-1)|H>>;
   R := simplify(M.R);
   QT := simplify(QT.Transpose(M));

   print();
   print('v' = v, 'H' = H, 'M' = M);
   print('R'[k] = R);

end do:
return [QT,R];
end proc:
```

```
> B := <<12,6,-4,0>|<-51,167,24,0>|<4,-68,-41,12>>;
QR := QR_decomposition(B):
```

$$B := \begin{bmatrix} 12 & -51 & 4 \\ 6 & 167 & -68 \\ -4 & 24 & -41 \\ 0 & 0 & 12 \end{bmatrix}$$

$$kolom = 1$$

$$R_0 = \begin{bmatrix} 12 & -51 & 4 \\ 6 & 167 & -68 \\ -4 & 24 & -41 \\ 0 & 0 & 12 \end{bmatrix}$$

$$v = \begin{bmatrix} 12 \\ 6 \\ -4 \\ 0 \end{bmatrix}, H = \begin{bmatrix} -\frac{6}{7} & -\frac{3}{7} & \frac{2}{7} & 0 \\ -\frac{3}{7} & \frac{82}{91} & \frac{6}{91} & 0 \\ \frac{2}{7} & \frac{6}{91} & \frac{87}{91} & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, M = \begin{bmatrix} -\frac{6}{7} & -\frac{3}{7} & \frac{2}{7} & 0 \\ -\frac{3}{7} & \frac{82}{91} & \frac{6}{91} & 0 \\ \frac{2}{7} & \frac{6}{91} & \frac{87}{91} & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$R_1 = \begin{bmatrix} -14 & -21 & 14 \\ 0 & \frac{2261}{13} & -\frac{854}{13} \\ 0 & \frac{252}{13} & -\frac{553}{13} \\ 0 & 0 & 12 \end{bmatrix}$$

$$kolom = 2$$

$$R_1 = \begin{bmatrix} -14 & -21 & 14 \\ 0 & \frac{2261}{13} & -\frac{854}{13} \\ 0 & \frac{252}{13} & -\frac{553}{13} \\ 0 & 0 & 12 \end{bmatrix}$$

$$v = \begin{bmatrix} \frac{2261}{13} \\ \frac{252}{13} \\ 0 \end{bmatrix}, H = \begin{bmatrix} -\frac{323}{325} & -\frac{36}{325} & 0 \\ -\frac{36}{325} & \frac{323}{325} & 0 \\ 0 & 0 & 1 \end{bmatrix}, M = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & -\frac{323}{325} & -\frac{36}{325} & 0 \\ 0 & -\frac{36}{325} & \frac{323}{325} & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$R_2 = \begin{bmatrix} -14 & -21 & 14 \\ 0 & -175 & 70 \\ 0 & 0 & -35 \\ 0 & 0 & 12 \end{bmatrix}$$

$$kolom = 3$$

$$R_2 = \begin{bmatrix} -14 & -21 & 14 \\ 0 & -175 & 70 \\ 0 & 0 & -35 \\ 0 & 0 & 12 \end{bmatrix}$$

$$v = \begin{bmatrix} -35 \\ 12 \end{bmatrix}, H = \begin{bmatrix} -\frac{35}{37} & \frac{12}{37} \\ \frac{12}{37} & \frac{35}{37} \end{bmatrix}, M = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -\frac{35}{37} & \frac{12}{37} \\ 0 & 0 & \frac{12}{37} & \frac{35}{37} \end{bmatrix}$$

$$R_3 = \begin{bmatrix} -14 & -21 & 14 \\ 0 & -175 & 70 \\ 0 & 0 & 37 \\ 0 & 0 & 0 \end{bmatrix} \qquad \textbf{(4.2.1)}$$

```
> Q := QR[1]:
  R := QR[2]:
  print('Q' = Q, 'R' = R, 'B' = B);
```

$$Q = \begin{bmatrix} -\frac{6}{7} & \frac{69}{175} & -\frac{58}{185} & \frac{696}{6475} \\ -\frac{3}{7} & -\frac{158}{175} & \frac{6}{185} & -\frac{72}{6475} \\ \frac{2}{7} & -\frac{6}{35} & -\frac{33}{37} & \frac{396}{1295} \\ 0 & 0 & \frac{12}{37} & \frac{35}{37} \end{bmatrix}, R = \begin{bmatrix} -14 & -21 & 14 \\ 0 & -175 & 70 \\ 0 & 0 & 37 \\ 0 & 0 & 0 \end{bmatrix}, B \qquad \textbf{(4.2.2)}$$

$$= \begin{bmatrix} 12 & -51 & 4 \\ 6 & 167 & -68 \\ -4 & 24 & -41 \\ 0 & 0 & 12 \end{bmatrix}$$

```
> Equal( Q.R, B );
```

**(4.2.3)**

$$true \qquad (4.2.3)$$

## 5.5. QR-decompositie en kleinste-kwadraten

We kunnen nu de QR-decompositie gebruiken om een overbepaald lineair stelsel op te lossen in de kleinste-kwadraten zin. Aangezien orthogonale matrices de norm bewaren vinden we

$$\|b - Ac\|_2 = \|Q^T b - Q^T Ac\|_2$$

Als we nu gebruiken dat $A = QR$ en $Q^T Q = \mathrm{I}$, vinden we bovendien dat

$$\|Q^T b - Q^T Ac\|_2 = \|Q^T b - Rc\|_2$$

Dit laatste stelsel $Rc = Q^T b$ is nu natuurlijk makkelijk op te lossen met behulp van achterwaartse substitutie aangezien $R$ een bovendriehoeksmatrix is.

```
> Q,R := QRDecomposition(A,fullspan):
  print('Q' = Q);
  print('R' = R);
```

$$Q = \Bigg[\Bigg[\frac{1}{4}, -\frac{3\sqrt{85}}{68}, \frac{5\sqrt{357}}{204}, -\frac{7\sqrt{62985}}{3876}, \frac{3\sqrt{29393}}{1292}, -\frac{11\sqrt{12597}}{3876}, \frac{\sqrt{1385670}}{3876}, 0, 0, 0, 0, 0, 0, 0, 0, 0\Bigg],$$

$$\Bigg[\frac{1}{4}, -\frac{13\sqrt{85}}{340}, \frac{\sqrt{357}}{68}, -\frac{7\sqrt{62985}}{19380}, -\frac{\sqrt{29393}}{1292}, \frac{11\sqrt{12597}}{3876}, -\frac{\sqrt{1385670}}{1615}, \frac{\sqrt{3233230}}{6460}, 0, 0, 0, 0, 0, 0, 0, 0\Bigg],$$

$$\Bigg[\frac{1}{4}, -\frac{11\sqrt{85}}{340}, \frac{3\sqrt{357}}{476}, \frac{11\sqrt{62985}}{19380}, -\frac{\sqrt{29393}}{532}, \frac{11\sqrt{12597}}{3876}, \frac{\sqrt{1385670}}{6460}, -\frac{9\sqrt{3233230}}{22610}, \frac{\sqrt{323323}}{2261}, 0, 0, 0, 0, 0, 0, 0\Bigg],$$

$$\Bigg[\frac{1}{4}, -\frac{9\sqrt{85}}{340}, -\frac{\sqrt{357}}{1428}, \frac{89\sqrt{62985}}{83980}, -\frac{201\sqrt{29393}}{117572}, \frac{11\sqrt{12597}}{16796}, \frac{37\sqrt{1385670}}{125970}, \frac{81\sqrt{3233230}}{587860}, -\frac{36\sqrt{323323}}{29393}, \frac{\sqrt{2431}}{221}, 0, 0, 0, 0, 0, 0\Bigg],$$

$$\Bigg[\frac{1}{4}, -\frac{7\sqrt{85}}{340}, -\frac{3\sqrt{357}}{476}, \frac{301\sqrt{62985}}{251940}, -\frac{101\sqrt{29393}}{117572}, -\frac{77\sqrt{12597}}{50388},$$

$$\frac{\sqrt{1385670}}{6460}, \frac{29\sqrt{3233230}}{146965}, \frac{33\sqrt{323323}}{58786}, -\frac{3\sqrt{2431}}{221}, \frac{\sqrt{7293}}{442}, 0, 0, 0, 0, 0\Bigg],$$

$$\Bigg[\frac{1}{4}, -\frac{\sqrt{85}}{68}, -\frac{5\sqrt{357}}{476}, \frac{53\sqrt{62985}}{50388}, \frac{23\sqrt{29393}}{117572}, -\frac{131\sqrt{12597}}{50388}, -\frac{8\sqrt{1385670}}{230945}, \frac{471\sqrt{3233230}}{6466460}, \frac{199\sqrt{323323}}{323323}, \frac{3\sqrt{2431}}{374}, -\frac{18\sqrt{7293}}{2431}, \frac{\sqrt{2145}}{286}, 0, 0, 0, 0\Bigg],$$

$$\Bigg[\frac{1}{4}, -\frac{3\sqrt{85}}{340}, -\frac{19\sqrt{357}}{1428}, \frac{179\sqrt{62985}}{251940}, \frac{129\sqrt{29393}}{117572}, -\frac{115\sqrt{12597}}{50388}, -\frac{5\sqrt{1385670}}{32604}, -\frac{3\sqrt{3233230}}{49742}, \frac{69\sqrt{323323}}{646646}, \frac{16\sqrt{2431}}{2431}, \frac{27\sqrt{7293}}{4862}, -\frac{9\sqrt{2145}}{715}, \frac{2\sqrt{2145}}{715}, 0, 0, 0\Bigg],$$

$$\Bigg[\frac{1}{4}, -\frac{\sqrt{85}}{340}, -\frac{\sqrt{357}}{68}, \frac{21\sqrt{62985}}{83980}, \frac{27\sqrt{29393}}{16796}, -\frac{15\sqrt{12597}}{16796}, -\frac{15\sqrt{1385670}}{92378}, -\frac{155\sqrt{3233230}}{1293292}, -\frac{6\sqrt{323323}}{19019}, -\frac{3\sqrt{2431}}{4862}, \frac{23\sqrt{7293}}{7293}, \frac{17\sqrt{2145}}{1430}, -\frac{8\sqrt{2145}}{715}, \frac{2\sqrt{429}}{429}, 0, 0\Bigg],$$

$$\Bigg[\frac{1}{4}, \frac{\sqrt{85}}{340}, -\frac{\sqrt{357}}{68}, -\frac{21\sqrt{62985}}{83980}, \frac{27\sqrt{29393}}{16796}, \frac{15\sqrt{12597}}{16796}, -\frac{15\sqrt{1385670}}{184756}, -\frac{30\sqrt{3233230}}{323323}, -\frac{10\sqrt{323323}}{24871}, -\frac{12\sqrt{2431}}{2431}, -\frac{4\sqrt{7293}}{2431}, \frac{8\sqrt{2145}}{2145}, \frac{19\sqrt{2145}}{1430}, -\frac{3\sqrt{429}}{143}, \frac{\sqrt{3003}}{858}, 0\Bigg],$$

$$\Bigg[\frac{1}{4}, \frac{3\sqrt{85}}{340}, -\frac{19\sqrt{357}}{1428}, -\frac{179\sqrt{62985}}{251940}, \frac{129\sqrt{29393}}{117572}, \frac{115\sqrt{12597}}{50388}, \frac{5\sqrt{1385670}}{138567}, -\frac{15\sqrt{3233230}}{1293292}, -\frac{60\sqrt{323323}}{323323}, -\frac{10\sqrt{2431}}{2431}, -\frac{8\sqrt{7293}}{2431}, -\frac{4\sqrt{2145}}{715}, \frac{\sqrt{2145}}{2145}, \frac{9\sqrt{429}}{286}, -\frac{6\sqrt{3003}}{1001}, \frac{\sqrt{231}}{462}\Bigg],$$

$$\Bigg[\frac{1}{4}, \frac{\sqrt{85}}{68}, -\frac{5\sqrt{357}}{476}, -\frac{53\sqrt{62985}}{50388}, \frac{23\sqrt{29393}}{117572}, \frac{131\sqrt{12597}}{50388},$$

$$\frac{23\sqrt{1385670}}{184756}, \frac{45\sqrt{3233230}}{646646}, \frac{45\sqrt{323323}}{323323}, 0, -\frac{10\sqrt{7293}}{7293}, -\frac{4\sqrt{2145}}{715},$$

$$-\frac{\sqrt{2145}}{130}, -\frac{4\sqrt{429}}{429}, \frac{23\sqrt{3003}}{2002}, -\frac{\sqrt{231}}{77}\Bigg],$$

$$\Bigg[\frac{1}{4}, \frac{7\sqrt{85}}{340}, -\frac{3\sqrt{357}}{476}, -\frac{301\sqrt{62985}}{251940}, -\frac{101\sqrt{29393}}{117572}, \frac{77\sqrt{12597}}{50388},$$

$$\frac{61\sqrt{1385670}}{461890}, \frac{639\sqrt{3233230}}{6466460}, \frac{108\sqrt{323323}}{323323}, \frac{9\sqrt{2431}}{2431}, \frac{4\sqrt{7293}}{2431},$$

$$\frac{2\sqrt{2145}}{2145}, -\frac{2\sqrt{2145}}{715}, -\frac{5\sqrt{429}}{286}, -\frac{25\sqrt{3003}}{3003}, \frac{5\sqrt{231}}{154}\Bigg],$$

$$\Bigg[\frac{1}{4}, \frac{9\sqrt{85}}{340}, -\frac{\sqrt{357}}{1428}, -\frac{89\sqrt{62985}}{83980}, -\frac{201\sqrt{29393}}{117572}, -\frac{11\sqrt{12597}}{16796},$$

$$\frac{11\sqrt{1385670}}{251940}, \frac{81\sqrt{3233230}}{1616615}, \frac{9\sqrt{323323}}{38038}, \frac{9\sqrt{2431}}{2431}, \frac{\sqrt{7293}}{374}, \frac{4\sqrt{2145}}{715},$$

$$\frac{23\sqrt{2145}}{4290}, \frac{\sqrt{429}}{143}, -\frac{5\sqrt{3003}}{2002}, -\frac{10\sqrt{231}}{231}\Bigg],$$

$$\Bigg[\frac{1}{4}, \frac{11\sqrt{85}}{340}, \frac{3\sqrt{357}}{476}, -\frac{11\sqrt{62985}}{19380}, -\frac{\sqrt{29393}}{532}, -\frac{11\sqrt{12597}}{3876},$$

$$-\frac{2\sqrt{1385670}}{20995}, -\frac{31\sqrt{3233230}}{587860}, -\frac{3\sqrt{323323}}{24871}, -\frac{3\sqrt{2431}}{4862}, \frac{2\sqrt{7293}}{7293},$$

$$\frac{3\sqrt{2145}}{1430}, \frac{3\sqrt{2145}}{715}, \frac{\sqrt{429}}{66}, \frac{8\sqrt{3003}}{1001}, \frac{5\sqrt{231}}{154}\Bigg],$$

$$\Bigg[\frac{1}{4}, \frac{13\sqrt{85}}{340}, \frac{\sqrt{357}}{68}, \frac{7\sqrt{62985}}{19380}, -\frac{\sqrt{29393}}{1292}, -\frac{11\sqrt{12597}}{3876},$$

$$-\frac{\sqrt{1385670}}{6460}, -\frac{33\sqrt{3233230}}{293930}, -\frac{23\sqrt{323323}}{58786}, -\frac{12\sqrt{2431}}{2431}, -\frac{15\sqrt{7293}}{4862},$$

$$-\frac{\sqrt{2145}}{165}, -\frac{9\sqrt{2145}}{1430}, -\frac{2\sqrt{429}}{143}, -\frac{29\sqrt{3003}}{6006}, -\frac{\sqrt{231}}{77}\Bigg],$$

$$\Bigg[\frac{1}{4}, \frac{3\sqrt{85}}{68}, \frac{5\sqrt{357}}{204}, \frac{7\sqrt{62985}}{3876}, \frac{3\sqrt{29393}}{1292}, \frac{11\sqrt{12597}}{3876}, \frac{\sqrt{1385670}}{9690},$$

$$\frac{3\sqrt{3233230}}{45220}, \frac{6\sqrt{323323}}{29393}, \frac{\sqrt{2431}}{442}, \frac{3\sqrt{7293}}{2431}, \frac{3\sqrt{2145}}{1430}, \frac{4\sqrt{2145}}{2145},$$

$$\frac{\sqrt{429}}{286}, \frac{\sqrt{3003}}{1001}, \frac{\sqrt{231}}{462}\Bigg]\Bigg]$$

$$R = \begin{bmatrix} 4 & 30 & 310 & 3600 & 44578 & 574800 \\ 0 & 2\sqrt{85} & 30\sqrt{85} & \frac{2068\sqrt{85}}{5} & 5658\sqrt{85} & 77876\sqrt{85} \\ 0 & 0 & 4\sqrt{357} & 90\sqrt{357} & \frac{10960\sqrt{357}}{7} & \frac{174750\sqrt{357}}{7} \\ 0 & 0 & 0 & \frac{6\sqrt{62985}}{5} & 36\sqrt{62985} & 758\sqrt{62985} \\ 0 & 0 & 0 & 0 & \frac{48\sqrt{29393}}{7} & \frac{1800\sqrt{29393}}{7} \\ 0 & 0 & 0 & 0 & 0 & 40\sqrt{12597} \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{4.3.1}$$

We bevestigen dat $Q$ inderdaad een orthogonale matrix is

```
> Transpose(Q).Q;
```

**(4.3.2)**

$$\left[\begin{array}{cccccccccccccccc}
1&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&1&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&1&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&1&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&1&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&1&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&1&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&1&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&1&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&1&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&1&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&1&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&1&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&1&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&1&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&1
\end{array}\right] \qquad \textbf{(4.3.2)}$$

We transformeren het overbepaald stelsel nu naar bovendriehoeksvorm (zonder het kleineste-kwadraten residu aan te passem!) door voor te vermenigvuldigen met de getransponeerde matrix $Q^T$.

```
> print('R' = R, 'c' = Vector(n+1,i->c[i-1]), 'Q^T*b' =
  Transpose(Q).b);
```

**(4.3.3)**

$$R = \begin{bmatrix} 4 & 30 & 310 & 3600 & 44578 & 574800 \\ 0 & 2\sqrt{85} & 30\sqrt{85} & \frac{2068\sqrt{85}}{5} & 5658\sqrt{85} & 77876\sqrt{85} \\ 0 & 0 & 4\sqrt{357} & 90\sqrt{357} & \frac{10960\sqrt{357}}{7} & \frac{174750\sqrt{357}}{7} \\ 0 & 0 & 0 & \frac{6\sqrt{62985}}{5} & 36\sqrt{62985} & 758\sqrt{62985} \\ 0 & 0 & 0 & 0 & \frac{48\sqrt{29393}}{7} & \frac{1800\sqrt{29393}}{7} \\ 0 & 0 & 0 & 0 & 0 & 40\sqrt{12597} \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}, c \tag{4.3.3}$$

$$
= \begin{bmatrix} c_0 \\ c_1 \\ c_2 \\ c_3 \\ c_4 \\ c_5 \end{bmatrix}, Q^T b = \begin{bmatrix} \frac{343}{8} \\ -\frac{1123\sqrt{85}}{680} \\ \frac{439\sqrt{357}}{952} \\ \frac{16589\sqrt{62985}}{503880} \\ -\frac{14765\sqrt{29393}}{235144} \\ \frac{2753\sqrt{12597}}{100776} \\ \frac{3\sqrt{1385670}}{17765} \\ -\frac{471\sqrt{3233230}}{12932920} \\ -\frac{\sqrt{323323}}{38038} \\ \frac{35\sqrt{2431}}{9724} \\ \frac{\sqrt{7293}}{7293} \\ -\frac{23\sqrt{2145}}{8580} \\ -\frac{17\sqrt{2145}}{2860} \\ -\frac{5\sqrt{429}}{1716} \\ \frac{25\sqrt{3003}}{12012} \\ \frac{\sqrt{231}}{308} \end{bmatrix}
$$

We kunnen nu dit stelsel oplossen aan de hand van achterwaartse substitutie.

```
> QTb := Transpose(Q).b:
  c := BackwardSubstitute(R[1..6,1..6],QTb[1..6]);
```

$$c := \begin{bmatrix} \frac{7713}{646} \\ \frac{12632833}{1007760} \\ -\frac{1982083}{403104} \\ \frac{9039}{14144} \\ -\frac{14015}{403104} \\ \frac{2753}{4031040} \end{bmatrix} \quad \textbf{(4.3.4)}$$

```
> p3 := plot(add(c[i]*X^(i-1),i=1..6),X=0..15,color=green,
  legend="Kleinste kwadraten graad 5",thickness=2):
  display(p2,p3,p1);
```




We merken direct op dat de kleinste-kwadraten veelterm niet oscilleert langs de randen, de graad is namelijk veel lager. Maar deze veelterm interpoleert nu natuurlijk niet meer: deze tracht zo dicht mogelijk bij de punten te komen, maar gaat er niet door.

Het theoretische residu is natuurlijk gelijk aan de laatste 10 rijen van de vector $Q^T b$ aangezien het matrix-vector product $Rc$ in deze rijen altijd gelijk is aan 0.

```
> residu := QTb - R.c;
  norm_residu := Norm(residu,2);
```

$$residu := \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ \frac{3\sqrt{1385670}}{17765} \\ -\frac{471\sqrt{3233230}}{12932920} \\ -\frac{\sqrt{323323}}{38038} \\ \frac{35\sqrt{2431}}{9724} \\ \frac{\sqrt{7293}}{7293} \\ -\frac{23\sqrt{2145}}{8580} \\ -\frac{17\sqrt{2145}}{2860} \\ -\frac{5\sqrt{429}}{1716} \\ \frac{25\sqrt{3003}}{12012} \\ \frac{\sqrt{231}}{308} \end{bmatrix}$$

$$rom\_residu := \frac{\sqrt{581230}}{1768} \tag{4.3.5}$$

De norm van dit residu is gelijk aan de norm van het originele residu. Maar de residuvectoren zijn natuurlijk niet gelijk!

```
> original_residu := b-A.c;
  norm_original_residu := Norm(original_residu,2);
```

$$original\_residu := \begin{bmatrix} \frac{39}{646} \\ -\frac{843}{5168} \\ \frac{205}{2584} \\ \frac{6887}{67184} \\ -\frac{107}{988} \\ \frac{1395}{67184} \\ \frac{2357}{33592} \\ \frac{3025}{67184} \\ -\frac{1737}{8398} \\ -\frac{5163}{67184} \\ \frac{7469}{33592} \\ \frac{265}{3536} \\ -\frac{155}{1292} \\ -\frac{157}{5168} \\ \frac{69}{2584} \\ \frac{1}{304} \end{bmatrix}$$

$$norm\_original\_residu := \frac{\sqrt{581230}}{1768} \qquad \textbf{(4.3.6)}$$

Aan de hand van de residuvector zien we nogmaals dat de kleinste-kwadraten oplossing een benadering is van de data en dus ook niet interpoleert: geen enkel van de componenten is namelijk gelijk aan 0.

# Veranderen van basis

Tot nu toe hebben we de monomiale basis gebruikt om de interpolerende en kleinste-kwadraten veelterm uit te rekenen. Dit leidt tot lineaire stelsels met een Vandermonde matrix als coëfficiëntenmatrix, dis befaamd zijn voor hun slechte conditionering (zoals we hierboven zien en ook tijdens de demo over LU-decompositie). De vraag is nu natuurlijk of we een andere basis kunne kiezen zodat het probleem veel beter geconditioneerd is? Het antwoord hierop is ja!

In plaats vande monomiale basis zullen we nu de kleinste-kwadraten veelterm uitrekenen in de Legendre basis. Deze veeltermen worden gedifineerd aan de hand van de volgende recursie:

$$(n+1)\,P_{n+1}(x) = (2\,n+1)\,x\,P_n(x) - n\,P_{n-1}(x)$$

waarvoor geldt dat

$$P_0(x) = 1 \quad \text{en} \quad P_1(x) = x$$

We zien duidelijk dat de Legendre veelterm $P_k(x)$ een veelterm is van graad $k$, en dus dat de lineaire combinatie

$$\sum_{k=0}^{n} c_k\,P_k(x)$$

dus ook een veelterm is van ten hoogste graad $n$. Dit betekent dat we de Legendre veeltermen dus ook kunnen gebruiken als de basisfuncties om de kleinste-kwadraten veelterm uit te rekenen.

```
> unassign('x'):
  for k from 0 to 5 do
  P[k] := simplify(LegendreP(k,x));
  end do;
```

$$P_0 := 1$$

$$P_1 := x$$

$$P_2 := -\frac{1}{2} + \frac{3\,x^2}{2}$$

$$P_3 := \frac{5}{2}\,x^3 - \frac{3}{2}\,x$$

$$P_4 := \frac{3}{8} + \frac{35}{8}\,x^4 - \frac{15}{4}\,x^2$$

$$P_5 := \frac{63}{8}\,x^5 - \frac{35}{4}\,x^3 + \frac{15}{8}\,x \qquad \textbf{(5.1)}$$

Voordat we gebruik gaan maken van de Legendre veeltermen, herschalen we de datapunten eerst naar een interessanter interval: namelijk [-1,1]. Deze herschaling zorgt op zich ook al voor een lager conditiegetal. Nu is onze x-data equidistant over het interval [0,15], dus om dit om te zetten naar [-1,1] delen we de punten door 15/2 en trekken er vervolgens 1 van af.

```
> x_rescaled := [seq(i*2/15-1,i=0..15)];
```

$$x\_rescaled := \left[-1, -\frac{13}{15}, -\frac{11}{15}, -\frac{3}{5}, -\frac{7}{15}, -\frac{1}{3}, -\frac{1}{5}, -\frac{1}{15}, \frac{1}{15}, \frac{1}{5}, \frac{1}{3}, \frac{7}{15}, \frac{3}{5}, \frac{11}{15}, \frac{13}{15}, 1\right] \tag{5.2}$$

Dit gaat ons natuurlijk uiteindelijk ook een kleinste-kwadraten benadering geven op het interval [-1,1] en niet op [0,15], dus we moeten het resultaat achteraf terug transformeren door de omgekeerde transformatie: 1 optellen en vermenigvuldigen met 15/2.

Heel dir proces komt natuurlijk overeen met het tranformeren van de basisfuncties, we kunnen namelijk deze herschaling rechtstreeks doorvoeren in de basisfuncties:

$P_k\left(\frac{2}{15}x-1\right)$ in plaats van $P_k(x)$

```
> unassign('x'):
  for k from 0 to 5 do
  P[k] := simplify(LegendreP(k,2/15*x-1));
  end do;
```

$$P_0 := 1$$

$$P_1 := \frac{2x}{15} - 1$$

$$P_2 := 1 + \frac{2}{75}x^2 - \frac{2}{5}x$$

$$P_3 := \frac{4}{675}x^3 - \frac{2}{15}x^2 + \frac{4}{5}x - 1$$

$$P_4 := 1 + \frac{14}{10125}x^4 - \frac{28}{675}x^3 + \frac{2}{5}x^2 - \frac{4}{3}x$$

$$P_5 := \frac{28}{84375}x^5 - \frac{14}{1125}x^4 + \frac{112}{675}x^3 - \frac{14}{15}x^2 + 2x - 1 \tag{5.3}$$

Aan de hand van deze basisfunctie herdefiniëren we het kleinste-kwadraten probleem als

$$d_0 P_0(x_j) + d_1 P_1(x_j) + d_2 P_2(x_j) + d_3 P_3(x_j) + d_4 P_4(x_j) + d_5 P_5(x_j) = y_j$$

voor $j = 0..15$.

```
> unassign('A','b'):
  n := 5:
```

```
> x := [seq(i,i=0..15)]:
  y := [12, 20, 22, 20, 16, 12, 17/2, 6, 9/2, 9/2, 11/2, 13/2,
  15/2, 17/2, 9, 9]:
  A := Matrix(16,n+1,(i,j)->LegendreP(j-1,x_rescaled[i]));
```

$$A := \begin{bmatrix} 1 & -1 & 1 & -1 & 1 & -1 \\ 1 & -\frac{13}{15} & \frac{47}{75} & -\frac{221}{675} & \frac{269}{10125} & \frac{689}{3125} \\ 1 & -\frac{11}{15} & \frac{23}{75} & \frac{77}{675} & -\frac{3811}{10125} & \frac{11407}{28125} \\ 1 & -\frac{3}{5} & \frac{1}{25} & \frac{9}{25} & -\frac{51}{125} & \frac{477}{3125} \\ 1 & -\frac{7}{15} & -\frac{13}{75} & \frac{301}{675} & -\frac{2371}{10125} & -\frac{4501}{28125} \\ 1 & -\frac{1}{3} & -\frac{1}{3} & \frac{11}{27} & \frac{1}{81} & -\frac{1}{3} \\ 1 & -\frac{1}{5} & -\frac{11}{25} & \frac{7}{25} & \frac{29}{125} & -\frac{961}{3125} \\ 1 & -\frac{1}{15} & -\frac{37}{75} & \frac{67}{675} & \frac{3629}{10125} & -\frac{3443}{28125} \\ 1 & \frac{1}{15} & -\frac{37}{75} & -\frac{67}{675} & \frac{3629}{10125} & \frac{3443}{28125} \\ 1 & \frac{1}{5} & -\frac{11}{25} & -\frac{7}{25} & \frac{29}{125} & \frac{961}{3125} \\ 1 & \frac{1}{3} & -\frac{1}{3} & -\frac{11}{27} & \frac{1}{81} & \frac{1}{3} \\ 1 & \frac{7}{15} & -\frac{13}{75} & -\frac{301}{675} & -\frac{2371}{10125} & \frac{4501}{28125} \\ 1 & \frac{3}{5} & \frac{1}{25} & -\frac{9}{25} & -\frac{51}{125} & -\frac{477}{3125} \\ 1 & \frac{11}{15} & \frac{23}{75} & -\frac{77}{675} & -\frac{3811}{10125} & -\frac{11407}{28125} \\ 1 & \frac{13}{15} & \frac{47}{75} & \frac{221}{675} & \frac{269}{10125} & -\frac{689}{3125} \\ 1 & 1 & 1 & 1 & 1 & 1 \end{bmatrix} \quad (5.4)$$

```
> b := Vector(16,i -> y[i]);
```

$$b := \begin{bmatrix} 12 \\ 20 \\ 22 \\ 20 \\ 16 \\ 12 \\ \frac{17}{2} \\ 6 \\ \frac{9}{2} \\ \frac{9}{2} \\ \frac{11}{2} \\ \frac{13}{2} \\ \frac{15}{2} \\ \frac{17}{2} \\ 9 \\ 9 \end{bmatrix} \tag{5.5}$$

```
> print( 'A' = A, 'd' = Vector(n+1,i->d[i-1]), 'b' = b );
```

$$A = \begin{bmatrix} 1 & -1 & 1 & -1 & 1 & -1 \\ 1 & -\frac{13}{15} & \frac{47}{75} & -\frac{221}{675} & \frac{269}{10125} & \frac{689}{3125} \\ 1 & -\frac{11}{15} & \frac{23}{75} & \frac{77}{675} & -\frac{3811}{10125} & \frac{11407}{28125} \\ 1 & -\frac{3}{5} & \frac{1}{25} & \frac{9}{25} & -\frac{51}{125} & \frac{477}{3125} \\ 1 & -\frac{7}{15} & -\frac{13}{75} & \frac{301}{675} & -\frac{2371}{10125} & -\frac{4501}{28125} \\ 1 & -\frac{1}{3} & -\frac{1}{3} & \frac{11}{27} & \frac{1}{81} & -\frac{1}{3} \\ 1 & -\frac{1}{5} & -\frac{11}{25} & \frac{7}{25} & \frac{29}{125} & -\frac{961}{3125} \\ 1 & -\frac{1}{15} & -\frac{37}{75} & \frac{67}{675} & \frac{3629}{10125} & -\frac{3443}{28125} \\ 1 & \frac{1}{15} & -\frac{37}{75} & -\frac{67}{675} & \frac{3629}{10125} & \frac{3443}{28125} \\ 1 & \frac{1}{5} & -\frac{11}{25} & -\frac{7}{25} & \frac{29}{125} & \frac{961}{3125} \\ 1 & \frac{1}{3} & -\frac{1}{3} & -\frac{11}{27} & \frac{1}{81} & \frac{1}{3} \\ 1 & \frac{7}{15} & -\frac{13}{75} & -\frac{301}{675} & -\frac{2371}{10125} & \frac{4501}{28125} \\ 1 & \frac{3}{5} & \frac{1}{25} & -\frac{9}{25} & -\frac{51}{125} & -\frac{477}{3125} \\ 1 & \frac{11}{15} & \frac{23}{75} & -\frac{77}{675} & -\frac{3811}{10125} & -\frac{11407}{28125} \\ 1 & \frac{13}{15} & \frac{47}{75} & \frac{221}{675} & \frac{269}{10125} & -\frac{689}{3125} \\ 1 & 1 & 1 & 1 & 1 & 1 \end{bmatrix}, d = \begin{bmatrix} d_0 \\ d_1 \\ d_2 \\ d_3 \\ d_4 \\ d_5 \end{bmatrix}, b = \begin{bmatrix} 12 \\ 20 \\ 22 \\ 20 \\ 16 \\ 12 \\ \frac{17}{2} \\ 6 \\ \frac{9}{2} \\ \frac{9}{2} \\ \frac{11}{2} \\ \frac{13}{2} \\ \frac{15}{2} \\ \frac{17}{2} \\ 9 \\ 9 \end{bmatrix} \quad (5.6)$$

```
> cond_rect(A);
```

$$3.09052062712583 \quad (5.7)$$

We vinden dat het conditiegetal slechts 3.09052062712583 is, wat vele malen lager is dan hetgeen dat we eerder hadden gevonden voor de monomiala basis (orde $10^6$). Het heeft dus uiteindelijk zeker geloond om een andere basis te kiezen.

Tenslotte lossen we nu ook dit stelsel eens op aan de hand van QR.

```
> Q,R := QRDecomposition(A,fullspan):
  QTb := Transpose(Q).b:
  d := BackwardSubstitute(R[1..6,1..6],QTb[1..6]);
```

$$d := \begin{bmatrix} \frac{11615121}{1074944} \\ -\frac{55613377}{7524608} \\ \frac{11823725}{1881152} \\ \frac{259425}{67184} \\ -\frac{49831875}{7524608} \\ \frac{15485625}{7524608} \end{bmatrix} \tag{5.8}$$

```
> unassign('x'):
  KK_legendre := collect(add(d[i+1]*P[i],i=0..5),x);
```

$$KK\_legendre := \frac{7713}{646} + \frac{12632833}{1007760}\,x - \frac{1982083}{403104}\,x^2 + \frac{9039}{14144}\,x^3 - \frac{14015}{403104}\,x^4 + \frac{2753}{4031040}\,x^5 \tag{5.9}$$

Dit is natuurlijk theoretisch gezien dezelfde veelterm als degene die we al hadden gevonden in de monomiale basis (de kleinste-kwadraten veelterm is namelijk uniek). Maar in een praktische toepassing, dus in floating-point aritmetiek en met ruis op de data, is het dus wel verstandig om een goede basis te kizen om zo de conditionering van het probleem onder controle te houden.

```
> KK_monomial := add(c[i+1]*x^i,i=0..5);
```

$$KK\_monomial := \frac{7713}{646} + \frac{12632833}{1007760}\,x - \frac{1982083}{403104}\,x^2 + \frac{9039}{14144}\,x^3 - \frac{14015}{403104}\,x^4 + \frac{2753}{4031040}\,x^5 \tag{5.10}$$